

# Automatizando a instalação de agentes em Servidores Windows

## INTRODUÇÃO:

---

Demonstrarei neste tutorial como automatizar a instalação de agentes Zabbix em Servidores Windows.

## PRÉ-REQUISITOS:

---

Um compartilhamento de rede;  
Os [binários](#) do Agente Zabbix nas versões 32 e 64 bits;  
Um servidor Zabbix previamente configurado e funcional.

## SCRIPT ZABBIX INSTALL.BAT:

---

Esse script é o responsável por realizar a tarefa.

```
REM #####  
REM # Oculta comentários #  
REM #####  
@echo off  
  
REM #####  
REM # Verifica existencia prévia do diretório #  
REM #####  
IF EXIST "c:\Zabbix\zabbix_agentd.exe" (exit) ELSE (echo "Iniciando Instalacao")  
  
REM #####  
REM # Exibe mensagem na tela #  
REM #####  
echo Copiando Arquivos  
  
REM #####  
REM # Mapeia a unidade de rede onde os binários estão disponíveis #  
REM # #
```

```
REM # Deve existir os diretórios win32 e win64, com os binários #
REM # # #
REM # Altere essa Informação #
REM #####
net use X: \\Seu_Dominio\Compartilhamento

REM #####
REM # Cria o diretório na máquina #
REM #####
mkdir c:\Zabbix

REM #####
REM # Verifica se o sistema é 32 ou 64 bits #
REM # #
REM # Realiza a cópia dos binários corretos para o sistema #
REM #####
IF EXIST "C:\Program Files (x86)" (
xcopy X:\win64\zabbix_agentd.exe c:\Zabbix
xcopy X:\win64\zabbix_get.exe c:\Zabbix
xcopy X:\win64\zabbix_sender.exe c:\Zabbix
)ELSE (
xcopy X:\win32\zabbix_agentd.exe c:\Zabbix
xcopy X:\win32\zabbix_get.exe c:\Zabbix
xcopy X:\win32\zabbix_sender.exe c:\Zabbix
)

REM #####
REM # Remove o mapeamento de rede #
REM #####
net use X: /delete

REM #####
REM # Exibe a mensagem na tela #
REM #####
echo Criando Arquivos de configuração

REM #####
REM # Gera o arquivo de configuração (zabbix_agentd.conf) na máquina #
REM # #
REM # Altere o nome do Servidor Zabbix #
REM #####
echo Server=Nome_DNS_Ou_IP_Do_Servidor_Zabbix > c:\Zabbix\zabbix_agentd.conf
```

```

echo Hostname=%COMPUTERNAME% >> c:\Zabbix\zabbix_agentd.conf
echo StartAgents=5 >> c:\Zabbix\zabbix_agentd.conf
echo DebugLevel=3 >> c:\Zabbix\zabbix_agentd.conf
echo LogFile=c:\Zabbix\zabbix_agentd.log >> c:\Zabbix\zabbix_agentd.conf
echo Timeout=3 >> c:\Zabbix\zabbix_agentd.conf

REM #####
REM # Exibe a mensagem na tela #
REM #####
echo Instalando o Serviço

REM #####
REM # Realiza a instalação do serviço #
REM # #
REM # Inicia o serviço #
REM #####
C:\Zabbix\zabbix_agentd.exe -i -c C:\Zabbix\zabbix_agentd.conf
C:\Zabbix\zabbix_agentd.exe -s -c C:\Zabbix\zabbix_agentd.conf

REM #####
REM # Encerra o Script #
REM #####
exit

```

## PREPARAÇÃO DO AMBIENTE:

Antes de usar o script precisamos realizar alguns passos simples.

Crie um diretório no servidor, baixe o binário (zabbix\_agents\_XXX.win.zip) para dentro deste diretório e descompacte o mesmo, será criada duas pastas "win32" e "win64". Compartilhe esse diretório na rede.

Altere as seguintes linhas no script:

```

net use X: \\Seu_Dominio\Compartilhamento
echo Server=Nome_DNS_Ou_IP_Do_Servidor_Zabbix > c:\Zabbix\zabbix_agentd.conf

```

## COMO USAR O SCRIPT:

---

Isso vai depender da estrutura da rede onde será usado, pode ser usado no momento da inicialização da máquina via Active Directory ou Samba. Pode ser executado o script Zabbix\_Install.bat manualmente em cada Servidor Windows a ser monitorado.

## CRIAR O HOST NO SERVIDOR ZABBIX:

---

Atenção nesse passo, pois o comando "**echo %COMPUTERNAME%**" retorna o nome do seu servidor em maiúsculo, como por exemplo "**INFOZABBIX**". Lembre-se de criar o host dessa maneira no Zabbix Server.

## REFERÊNCIAS:

---

[Fórum Zabbix](#)

**Criado por:** [André Déo](#)